The $\lambda\mu$-calculus

Cristina Matache Victor B. F. Gomes Dominic P. Mulligan

March 17, 2025

Contents

Abstract

The propositions-as-types correspondence is ordinarily presented as linking the metatheory of typed λ-calculi and the proof theory of intuitionistic logic. Griffin [2] observed that this correspondence could be extended to classical logic through the use of control operators. This observation set off a flurry of further research, leading to the development of Parigots $\lambda\mu$-calculus [4]. In this work, we formalise $\lambda\mu$-calculus in Isabelle/HOL and prove several metatheoretical properties such as type preservation and progress.

1 The $\lambda\mu$-calculus

More examples, as well as a call-by-value programming language built on top of our formalisation, can be found in an associated Bitbucket repository [3].

theory *Syntax*
imports *Main*
begin

1.1 Syntax

datatype *type* =
  Iota
| *Fun type type* (**infixr** ‹→› *200*)

To deal with α-equivalence, we use De Bruijn's nameless representation wherein each bound variable is represented by a natural number, its index, that denotes the number of binders that must be traversed to arrive at the one that binds the given variable. Each free variable has an index that points into the top-level context, not enclosed in any abstractions.

datatype *trm* =
    LVar nat (‹‘-› [*100*] *100*)
| *Lbd type trm* (‹λ-:-› [*0*, *60*] *60*)
| *App trm trm* (**infix** ‹°› *60*)
| *Mu type cmd* (‹μ-:-› [*0*, *60*] *60*)
and *cmd* =
    MVar nat trm (‹<->-› [*0*, *60*] *60*)

datatype *ctxt* =
  ContextEmpty (‹◊›) |
  ContextApp ctxt trm (**infixl** ‹•› *75*)

primrec *ctxt-app* :: *ctxt* ⇒ *ctxt* ⇒ *ctxt* (**infix** ‹.› *60*) **where**
  ◊ . *F* = *F* |
  (*E* • *t*) . *F* = (*E* . *F*) • *t*

fun *is-val* :: *trm* ⇒ *bool* **where**
  is-val (λ *T* : *v*) = *True* |
  is-val - = *False*

end

1.2 Types

theory *Types*
  imports *Syntax*
begin

We implement typing environments as (total) functions from natural numbers to types, following the approach of Stefan Berghofer in his formalisation of the simply typed λ-calculus in the Isabelle/HOL library. An empty typing environment may be represented by an arbitrary function of the correct type as it will never be queried when a typing judgement is valid. We split typing environments, dedicating one environment to λ-variables

and another to μ-variables, and use Γ and Δ to range over the former and latter, respectively.

From src/HOL/Proofs/LambdaType.thy

definition
shift :: $(nat \Rightarrow 'a) \Rightarrow nat \Rightarrow 'a \Rightarrow nat \Rightarrow 'a$ $(\langle\text{-}\langle\text{-:-}\rangle\rangle\ [90,\ 0,\ 0]\ 91)$
where
$e\langle i{:}a\rangle = (\lambda j.\ \textit{if}\ j < i\ \textit{then}\ e\ j\ \textit{else if}\ j = i\ \textit{then}\ a\ \textit{else}\ e\ (j-1))$

lemma *shift-eq* [*simp*]: $i = j \Longrightarrow (e\langle i{:}T\rangle)\ j = T$
by (*simp add*: *shift-def*)

lemma *shift-gt* [*simp*]: $j < i \Longrightarrow (e\langle i{:}T\rangle)\ j = e\ j$
by (*simp add*: *shift-def*)

lemma *shift-lt* [*simp*]: $i < j \Longrightarrow (e\langle i{:}T\rangle)\ j = e\ (j - 1)$
by (*simp add*: *shift-def*)

lemma *shift-commute* [*simp*]: $e\langle i{:}U\rangle\langle 0{:}T\rangle = e\langle 0{:}T\rangle\langle Suc\ i{:}U\rangle$
by (*rule ext*) (*simp-all add*: *shift-def split*: *nat.split*, *force*)

inductive *typing-trm* :: $(nat \Rightarrow type) \Rightarrow (nat \Rightarrow type) \Rightarrow trm \Rightarrow type \Rightarrow bool$
$(\langle\text{-}\ ,\ \text{-}\vdash_T\ \text{-}:\ \text{-}\rangle\ [50,\ 50,\ 50,\ 50]\ 50)$
and *typing-cmd* :: $(nat \Rightarrow type) \Rightarrow (nat \Rightarrow type) \Rightarrow cmd \Rightarrow bool$
$(\langle\text{-}\ ,\ \text{-}\vdash_C\ \text{-}\rangle\ [50,\ 50,\ 50]\ 50)$
where
var [*intro*!]: $[\![\ \Gamma\ x = T\ ]\!] \Longrightarrow \Gamma, \Delta \vdash_T\ `x : T\ |$
app [*intro*!]: $[\![\ \Gamma, \Delta \vdash_T t : (T1 \to T2); \Gamma, \Delta \vdash_T s : T1\ ]\!] \Longrightarrow \Gamma, \Delta \vdash_T (t\circ s) : T2\ |$
lambda [*intro*!]: $[\![\ \Gamma\langle 0{:}T1\rangle, \Delta \vdash_T t : T2\ ]\!] \Longrightarrow \Gamma, \Delta \vdash_T (\lambda\ T1 : t) : (T1 \to T2)\ |$
activate [*intro*!]: $[\![\ \Gamma, \Delta\langle 0{:}T\rangle \vdash_C c\ ]\!] \Longrightarrow \Gamma, \Delta \vdash_T (\mu\ T : c) : T\ |$
passivate [*intro*!]: $[\![\ \Gamma, \Delta \vdash_T t : T; \Delta\ x = T]\!] \Longrightarrow \Gamma, \Delta \vdash_C (<x>\ t)$

inductive-cases *typing-elims* [*elim*!]:
$\Gamma, \Delta \vdash_T\ `x : T$
$\Gamma, \Delta \vdash_T t\circ s : T$
$\Gamma, \Delta \vdash_T \lambda\ T1 : t : T$
$\Gamma, \Delta \vdash_T \mu\ T1 : t : T$
$\Gamma, \Delta \vdash_C <x>\ t$

inductive-cases *type-arrow-elim*:
$\Gamma\ , \Delta \vdash_T t : T1 \to T2$

lemma *uniqueness*:
$\Gamma, \Delta \vdash_T t : T1 \Longrightarrow \Gamma, \Delta \vdash_T t : T2 \Longrightarrow T1 = T2$
$\Gamma, \Delta \vdash_C c \Longrightarrow \Gamma, \Delta \vdash_C c$
apply(*induct arbitrary*: *T2 rule*: *typing-trm-typing-cmd.inducts*)

using *typing-cmd.cases* **apply** *blast+*
done

end
theory *DeBruijn*
imports *Syntax*
begin

1.3 De Bruijn indices

Functions to find the free λ and μ variables in an expression.

primrec *flv-trm* :: *trm* ⇒ *nat* ⇒ *nat set*
and *flv-cmd* :: *cmd* ⇒ *nat* ⇒ *nat set*
where
flv-trm (‘*i*) *k* = (*if i*≥*k then* {*i*−*k*} *else* {})
| *flv-trm* (λ *T* : *t*) *k* = *flv-trm t* (*k*+*1*)
| *flv-trm* (*s*°*t*) *k* = (*flv-trm s k*) ∪ (*flv-trm t k*)
| *flv-trm* (μ *T* : *c*) *k* = *flv-cmd c k*
| *flv-cmd* (<*i*> *t*) *k* = *flv-trm t k*

primrec *fmv-trm* :: *trm* ⇒ *nat* ⇒ *nat set*
and *fmv-cmd* :: *cmd* ⇒ *nat* ⇒ *nat set*
where
fmv-trm (‘*i*) *k* = {}
| *fmv-trm* (λ *T* : *t*) *k* = *fmv-trm t k*
| *fmv-trm* (*s*°*t*) *k* = (*fmv-trm s k*) ∪ (*fmv-trm t k*)
| *fmv-trm* (μ *T* : *c*) *k* = *fmv-cmd c* (*k*+*1*)
| *fmv-cmd* (<*i*> *t*) *k* = (*if i*≥*k then* {*i*−*k*} ∪ (*fmv-trm t k*) *else* (*fmv-trm t k*))

abbreviation *lambda-closed* :: - **where**
lambda-closed t ≡ *flv-trm t 0* = {}

abbreviation *lambda-closedC* :: - **where**
lambda-closedC c ≡ *flv-cmd c 0* = {}

Free variables in a context.

primrec *fmv-ctxt* :: *ctxt* ⇒ *nat* ⇒ *nat set* **where**
fmv-ctxt ◊ *k* = {}
| *fmv-ctxt* (*E* • *t*) *k* = (*fmv-ctxt E k*) ∪ (*fmv-trm t k*)

Shift free λ and μ variables in terms and commands to make substitution capture avoiding.

primrec
liftL-trm :: [*trm*, *nat*] ⇒ *trm* **and**
liftL-cmd :: [*cmd*, *nat*] ⇒ *cmd*
where
liftL-trm (‘*i*) *k* = (*if i*<*k then* ‘*i else* ‘(*i*+*1*)) |

liftL-trm (λ T : t) k = λ T : (*liftL-trm* t (k+*1*)) |
liftL-trm (s $\circ$ t) k = *liftL-trm* s k $\circ$ *liftL-trm* t k |
liftL-trm (μ T : c) k = μ T : (*liftL-cmd* c k) |
liftL-cmd (<i> t) k = <i> (*liftL-trm* t k)

primrec
liftM-trm :: [*trm*, *nat*] $\Rightarrow$ *trm* **and**
liftM-cmd :: [*cmd*, *nat*] $\Rightarrow$ *cmd*
where
liftM-trm (‘i) k = ‘i |
liftM-trm (λ T : t) k = λ T : (*liftM-trm* t k) |
liftM-trm (s $\circ$ t) k = *liftM-trm* s k $\circ$ *liftM-trm* t k |
liftM-trm (μ T : c) k = μ T : (*liftM-cmd* c (k+*1*)) |
liftM-cmd (<i> t) k =
(*if* i<k *then* (<i> (*liftM-trm* t k)) *else* (<i+*1*> (*liftM-trm* t k)))

Shift free λ and μ variables in contexts to make structural substitution capture avoiding.

primrec *liftL-ctxt* :: *ctxt* $\Rightarrow$ *nat* $\Rightarrow$ *ctxt* **where**
liftL-ctxt $\Diamond$ n = $\Diamond$ |
liftL-ctxt (E $\bullet$ t) n = (*liftL-ctxt* E n) $\bullet$ (*liftL-trm* t n)

primrec *liftM-ctxt* :: *ctxt* $\Rightarrow$ *nat* $\Rightarrow$ *ctxt* **where**
liftM-ctxt $\Diamond$ n = $\Diamond$ |
liftM-ctxt (E $\bullet$ t) n = (*liftM-ctxt* E n) $\bullet$ (*liftM-trm* t n)

A function to decrement the indices of free μ-variables when a μ surrounding the expression disappears as a result of a reduction

primrec
dropM-trm :: [*trm*, *nat*] $\Rightarrow$ *trm* **and**
dropM-cmd :: [*cmd*, *nat*] $\Rightarrow$ *cmd*
where
dropM-trm (‘i) k = ‘i
| *dropM-trm* (λ T : t) k = λ T : (*dropM-trm* t k)
| *dropM-trm* (s $\circ$ t) k = (*dropM-trm* s k) $\circ$ (*dropM-trm* t k)
| *dropM-trm* (μ T : c) k = μ T : (*dropM-cmd* c (k+*1*))
| *dropM-cmd* (<i> t) k =
(*if* i>k *then* (<i−*1*> (*dropM-trm* t k)) *else* (<i> (*dropM-trm* t k)))

lemma *fmv-liftL*:
$\beta \notin$ *fmv-trm* t $n \Longrightarrow \beta \notin$ *fmv-trm* (*liftL-trm* t m) n
$\beta \notin$ *fmv-cmd* c $n \Longrightarrow \beta \notin$ *fmv-cmd* (*liftL-cmd* c m) n
by(*induct* t **and** c *arbitrary*: m n **and** m n) *auto*

lemma *fmv-liftL-ctxt*:
$\beta \notin$ *fmv-ctxt* E $m \Longrightarrow \beta \notin$ *fmv-ctxt* (*liftL-ctxt* E n) m
by(*induct* E) (*fastforce simp add*: *fmv-liftL*)+

lemma *fmv-suc*:
β ∉ fmv-cmd c (Suc n) ⟹ (Suc β) ∉ fmv-cmd c n
β ∉ fmv-trm t (Suc n) ⟹ (Suc β) ∉ fmv-trm t n
proof (*induct c* **and** *t arbitrary*: *n* **and** *n*)
case (*MVar x1 x2*)
then show *?case*
by *clarsimp* (*metis UnI1 UnI2 diff-Suc-1 diff-Suc-eq-diff-pred diff-commute diff-is-0-eq nat.simps*(*3*) *not-less-eq-eq singletonI*)
qed (*force*)+

lemma *flv-drop*:
flv-trm t k = {} ⟶ flv-trm (*dropM-trm t j*) *k = {}*
flv-cmd c k = {} ⟶ flv-cmd (*dropM-cmd c j*) *k = {}*
by (*induct t* **and** *c arbitrary*: *j k* **and** *j k*) *clarsimp*+

end

1.4 Logical and structural substitution

theory *Substitution*
imports *DeBruijn*
begin

primrec
subst-trm :: [*trm*, *trm*, *nat*] ⇒ *trm* (‹-[-'/-]^T› [*300*, *0*, *0*] *300*) **and**
subst-cmd :: [*cmd*, *trm*, *nat*] ⇒ *cmd* (‹-[-'/-]^C› [*300*, *0*, *0*] *300*)
where
subst-LVar: (‘*i*)[*s*/*k*]^T =
(*if k < i then* ‘(*i−1*) *else if k = i then s else* (‘*i*))
| *subst-Lbd*: (λ *T* : *t*)[*s*/*k*]^T = λ *T* : (*t*[(*liftL-trm s 0*) / *k+1*]^T)
| *subst-App*: (*t* ° *u*)[*s*/*k*]^T = *t*[*s*/*k*]^T ° *u*[*s*/*k*]^T
| *subst-Mu*: (μ *T* : *c*)[*s*/*k*]^T = μ *T* : (*c*[(*liftM-trm s 0*) / *k*]^C)
| *subst-MVar*: (<*i*> *t*)[*s*/*k*]^C = <*i*> (*t*[*s*/*k*]^T)

Substituting a term for the hole in a context.

primrec *ctxt-subst* :: *ctxt* ⇒ *trm* ⇒ *trm* **where**
ctxt-subst ◊ *s* = *s* |
ctxt-subst (*E* • *t*) *s* = (*ctxt-subst E s*)° *t*

lemma *ctxt-app-subst*:
shows *ctxt-subst E* (*ctxt-subst F t*) = *ctxt-subst* (*E* . *F*) *t*
by (*induction E*, *auto*)

The structural substitution is based on Geuvers and al. [1].

primrec
struct-subst-trm :: [*trm*, *nat*, *nat*, *ctxt*] ⇒ *trm* (‹-[-=- -]^T› [*300*, *0*, *0*, *0*] *300*) **and**
struct-subst-cmd :: [*cmd*, *nat*, *nat*, *ctxt*] ⇒ *cmd* (‹-[-=- -]^C› [*300*, *0*, *0*, *0*] *300*)

where
struct-LVar: $(`i)[j{=}k\ E]^T = (`i)$ |
struct-Lbd: $(\lambda\ T : t)[j{=}k\ E]^T = (\lambda\ T : (t[j{=}k\ (liftL\text{-}ctxt\ E\ 0)]^T))$ |
struct-App: $(t \circ s)[j{=}k\ E]^T = (t[j{=}k\ E]^T) \circ (s[j{=}k\ E]^T)$ |
struct-Mu: $(\mu\ T : c)[j{=}k\ E]^T = \mu\ T : (c[(j{+}1){=}(k{+}1)\ (liftM\text{-}ctxt\ E\ 0)]^C)$ |
struct-MVar: $(<i>\ t)[j{=}k\ E]^C =$
$(if\ i{=}j\ then\ (<k>\ (ctxt\text{-}subst\ E\ (t[j{=}k\ E]^T)))$
$else\ (if\ j{<}i \wedge i{\leq}k\ then\ (<i{-}1>\ (t[j{=}k\ E]^T))$
$else\ (if\ k{\leq}i \wedge i{<}j\ then\ (<i{+}1>\ (t[j{=}k\ E]^T))$
$else\ (<i>\ (t[j{=}k\ E]^T)))))$

Lifting of lambda and mu variables commute with each other

lemma *liftLM-comm*:
$liftL\text{-}trm\ (liftM\text{-}trm\ t\ n)\ m = liftM\text{-}trm\ (liftL\text{-}trm\ t\ m)\ n$
$liftL\text{-}cmd\ (liftM\text{-}cmd\ c\ n)\ m = liftM\text{-}cmd\ (liftL\text{-}cmd\ c\ m)\ n$
by(*induct t* **and** *c arbitrary*: *n m* **and** *n m*) *auto*

lemma *liftLM-comm-ctxt*:
$liftL\text{-}ctxt\ (liftM\text{-}ctxt\ E\ n)\ m = liftM\text{-}ctxt\ (liftL\text{-}ctxt\ E\ m)\ n$
by(*induct E arbitrary*: *n m*, *auto simp add*: *liftLM-comm*)

Lifting of μ-variables (almost) commutes.

lemma *liftMM-comm*:
$n{\geq}m \Longrightarrow liftM\text{-}trm\ (liftM\text{-}trm\ t\ n)\ m = liftM\text{-}trm\ (liftM\text{-}trm\ t\ m)\ (Suc\ n)$
$n{\geq}m \Longrightarrow liftM\text{-}cmd\ (liftM\text{-}cmd\ c\ n)\ m = liftM\text{-}cmd\ (liftM\text{-}cmd\ c\ m)\ (Suc\ n)$
by(*induct t* **and** *c arbitrary*: *n m* **and** *n m*) *auto*

lemma *liftMM-comm-ctxt*:
$liftM\text{-}ctxt\ (liftM\text{-}ctxt\ E\ n)\ 0 = liftM\text{-}ctxt\ (liftM\text{-}ctxt\ E\ 0)\ (n{+}1)$
by(*induct E arbitrary*: *n*, *auto simp add*: *liftMM-comm*)

If a μ variable i doesn't occur in a term or a context, then these remain the same after structural substitution of variable i.

lemma *liftM-struct-subst*:
$liftM\text{-}trm\ t\ i[i{=}i\ F]^T = liftM\text{-}trm\ t\ i$
$liftM\text{-}cmd\ c\ i[i{=}i\ F]^C = liftM\text{-}cmd\ c\ i$
by(*induct t* **and** *c arbitrary*: *i F* **and** *i F*) *auto*

lemma *liftM-ctxt-struct-subst*:
$(ctxt\text{-}subst\ (liftM\text{-}ctxt\ E\ i)\ t)[i{=}i\ F]^T = ctxt\text{-}subst\ (liftM\text{-}ctxt\ E\ i)\ (t[i{=}i\ F]^T)$
by(*induct E arbitrary*: *i t F*; *force simp add*: *liftM-struct-subst*)

end

1.5 Reduction relation

theory *Reduction*
imports *Substitution*
begin

inductive *red-term* :: [*trm*, *trm*] ⇒ *bool* (**infixl** ‹⟶› *50*)
and *red-cmd* :: [*cmd*, *cmd*] ⇒ *bool* (**infixl** ‹$_C$⟶› *50*)
where
beta [*intro*]: $(\lambda\ T : t)\circ r \longrightarrow t[r/0]^T$ |
struct [*intro*]: $(\mu\ (T1 \to T2) : c)\circ s \longrightarrow \mu\ T2 : (c[0 = 0\ (\Diamond \bullet (\mathit{liftM\text{-}trm}\ s\ 0))]^C)$ |
rename [*intro*]: $[\![\ 0 \notin (\mathit{fmv\text{-}trm}\ t\ 0)\ ]\!] \Longrightarrow (\mu\ T : (<0> t)) \longrightarrow \mathit{dropM\text{-}trm}\ t\ 0$ |
mueta [*intro*]: $<i>\ (\mu\ T : c)\ {}_C\!\longrightarrow (\mathit{dropM\text{-}cmd}\ (c[0 = i\ \Diamond]^C)\ i)$ |

lambda [*intro*]: $[\![\ s \longrightarrow t\ ]\!] \Longrightarrow (\lambda\ T : s) \longrightarrow (\lambda\ T : t)$ |
appL [*intro*]: $[\![\ s \longrightarrow u\ ]\!] \Longrightarrow (s\circ t) \longrightarrow (u\circ t)$ |
appR [*intro*]: $[\![\ t \longrightarrow u\ ]\!] \Longrightarrow (s\circ t) \longrightarrow (s\circ u)$ |
mu [*intro*]: $[\![\ c\ {}_C\!\longrightarrow d\ ]\!] \Longrightarrow (\mu\ T : c) \longrightarrow (\mu\ T : d)$ |
cmd [*intro*]: $[\![\ t \longrightarrow s\ ]\!] \Longrightarrow (<i>\ t)\ {}_C\!\longrightarrow (<i>\ s)$

inductive-cases *redE* [*elim*]:
$`i \longrightarrow s$
$(\lambda\ T : t) \longrightarrow s$
$s\circ t \longrightarrow u$
$(\mu\ T : c) \longrightarrow t$
$<i>\ t\ {}_C\!\longrightarrow c$

Reflexive transitive closure

inductive *beta-rtc-term* :: [*trm*, *trm*] ⇒ *bool* (**infixl** ‹⟶*› *50*)
where
refl-term [*iff*]: $s \longrightarrow^* s$ |
step-term: $[\![s \longrightarrow t;\ t \longrightarrow^* u]\!] \Longrightarrow s \longrightarrow^* u$

lemma *step-term2*: $[\![s \longrightarrow^* t;\ t \longrightarrow u]\!] \Longrightarrow s \longrightarrow^* u$
by (*induct rule*: *beta-rtc-term.induct*) (*auto intro*: *step-term*)

inductive *beta-rtc-command* :: [*cmd*, *cmd*] ⇒ *bool* (**infixl** ‹$_C$⟶*› *50*) **where**
refl-command [*iff*]: $c\ {}_C\!\longrightarrow^* c$ |
step-command: $c\ {}_C\!\longrightarrow d \Longrightarrow d\ {}_C\!\longrightarrow^* e \Longrightarrow c\ {}_C\!\longrightarrow^* e$

The beta reduction relation is included in the reflexive transitive closure.

lemma *rtc-term-incl* [*intro*]: $s \longrightarrow t \Longrightarrow s \longrightarrow^* t$
by (*meson beta-rtc-term.simps*)

lemma [*intro*]: $c\ {}_C\!\longrightarrow\ d \Longrightarrow c\ {}_C\!\longrightarrow^*\ d$
by(*auto intro*: *step-command*)

Proof that the reflexive transitive closure as defined above is transitive.

lemma *rtc-term-trans* [*intro*]: $s \longrightarrow^*\ t \Longrightarrow t \longrightarrow^*\ u \Longrightarrow s \longrightarrow^*\ u$
by(*induction rule*: *beta-rtc-term.induct*) (*auto simp add*: *step-term*)

lemma *rtc-command-trans*[*intro*]: $c\ {}_C\!\longrightarrow^*\ d \Longrightarrow d\ {}_C\!\longrightarrow^*\ e \Longrightarrow c\ {}_C\!\longrightarrow^*\ e$
by(*induction rule*: *beta-rtc-command.induct*) (*auto simp add*: *step-command*)

Congruence rules for the reflexive transitive closure.

lemma *rtc-lambda*: $s \longrightarrow^*\ t \Longrightarrow (\lambda\ T : s) \longrightarrow^*\ (\lambda\ T : t)$
apply(*induction rule*: *beta-rtc-term.induct*)
by *force* (*meson red-term-red-cmd.lambda step-term*)

lemma *rtc-appL*: $s \longrightarrow^*\ u \Longrightarrow (s \circ t) \longrightarrow^*\ (u \circ t)$
apply(*induction rule*: *beta-rtc-term.induct*)
by *force* (*meson appL step-term*)

end

1.6 Contextual typing

theory *ContextFacts*
imports
Reduction
Types
begin

Naturally, we may wonder when instantiating the hole in a context is type-preserving. To assess this, we define a typing judgement for contexts.

inductive *typing-ctxt* :: (*nat* ⇒ *type*) ⇒ (*nat* ⇒ *type*) ⇒ *ctxt* ⇒ *type* ⇒ *type* ⇒ *bool*
$(\langle\text{-}\ ,\ \text{-} \vdash_{ctxt} \text{-} : \text{-} \Leftarrow \text{-}\rangle\ [50,\ 50,\ 50,\ 50,\ 50]\ 50)$
where
type-ctxtEmpty [*intro*!]: $\Gamma,\ \Delta \vdash_{ctxt} \Diamond : T \Leftarrow T\ |$
type-ctxtApp [*intro*!]: $[\![\ \Gamma,\ \Delta \vdash_{ctxt} E : (T1 \rightarrow T2) \Leftarrow U;\ \Gamma,\ \Delta \vdash_T t : T1\ ]\!] \Longrightarrow \Gamma,\ \Delta \vdash_{ctxt} (E \bullet t) : T2 \Leftarrow U$

inductive-cases *typing-ctxt-elims* [*elim*!]:
$\Gamma,\ \Delta \vdash_{ctxt} \Diamond : T \Leftarrow T$
$\Gamma,\ \Delta \vdash_{ctxt} (E \bullet t) : T \Leftarrow U$

lemma *typing-ctxt-correct1*:

shows $\Gamma,\ \Delta \vdash_T (\mathit{ctxt\text{-}subst}\ E\ r) : T \implies \exists\, U.\ (\Gamma,\ \Delta \vdash_T r : U \wedge \Gamma,\ \Delta \vdash_{ctxt} E : T \Leftarrow U)$
by(*induction E arbitrary*: $\Gamma\ \Delta\ T\ r$; *force*)

lemma *typing-ctxt-correct2*:
shows $\Gamma,\ \Delta \vdash_{ctxt} E : T \Leftarrow U \implies \Gamma,\ \Delta \vdash_T r : U \implies \Gamma,\ \Delta \vdash_T (\mathit{ctxt\text{-}subst}\ E\ r) : T$
by(*induction arbitrary*: *r rule*: *typing-ctxt.induct*) *auto*

lemma *ctxt-subst-basecase*:
$\forall\, n.\ c[n = n\ \Diamond]^C = c$
$\forall\, n.\ t[n = n\ \Diamond]^T = t$
by(*induct c* **and** *t*) (*auto*)

lemma *ctxt-subst-caseApp*:
$\forall\, n\ E\ s.\ (c[n{=}n\ (\mathit{liftM\text{-}ctxt}\ E\ n)]^C)[n{=}n\ (\Diamond \bullet (\mathit{liftM\text{-}trm}\ s\ n))]^C = c[n{=}n\ ((\mathit{liftM\text{-}ctxt}\ E\ n) \bullet (\mathit{liftM\text{-}trm}\ s\ n))]^C$
$\forall\, n\ E\ s.\ (t[n{=}n\ (\mathit{liftM\text{-}ctxt}\ E\ n)]^T)[n{=}n\ (\Diamond \bullet (\mathit{liftM\text{-}trm}\ s\ n))]^T = t[n{=}n\ ((\mathit{liftM\text{-}ctxt}\ E\ n) \bullet (\mathit{liftM\text{-}trm}\ s\ n))]^T$
by (*induction c* **and** *t*) (*auto simp add*: *liftLM-comm-ctxt liftLM-comm liftMM-comm-ctxt liftMM-comm liftM-ctxt-struct-subst*)

lemma *ctxt-subst*:
assumes $\Gamma,\ \Delta \vdash_{ctxt} E : U \Leftarrow T$
shows $(\mathit{ctxt\text{-}subst}\ E\ (\mu\ T : c)) \longrightarrow^* \mu\ U : (c[0 = 0\ (\mathit{liftM\text{-}ctxt}\ E\ 0)]^C)$
using *assms* **proof**(*induct E arbitrary*: $U\ T\ \Gamma\ \Delta\ c$)
case *ContextEmpty*
have *ctxtEmpty-inv*: $\Gamma,\ \Delta \vdash_{ctxt} \Diamond : U \Leftarrow T \implies U = T$
by(*cases* $\Gamma\ \Delta\ \Diamond$ *rule*: *typing-ctxt.cases*, *fastforce*, *simp*)
show *?case*
using *ContextEmpty* **by** (*clarsimp dest*!: *ctxtEmpty-inv simp*: *ctxt-subst-basecase*)
next
case (*ContextApp E x*)
then show *?case*
by *clarsimp* (*rule rtc-term-trans*, *rule rtc-appL*, *assumption*, *rule step-term*, *force*, *clarsimp simp add*: *ctxt-subst-caseApp*(*1*))
qed

end

1.7 Type preservation

theory *TypePreservation*
imports
ContextFacts
begin

Shifting lambda variables preserves well-typedness.

lemma *liftL-type*:
$\Gamma, \Delta \vdash_T t : T \Longrightarrow \forall k.\ \Gamma\langle k{:}U\rangle, \Delta \vdash_T (\mathit{liftL\text{-}trm}\ t\ k) : T$
$\Gamma, \Delta \vdash_C c \Longrightarrow \forall k.\ \Gamma\langle k{:}U\rangle, \Delta \vdash_C (\mathit{liftL\text{-}cmd}\ c\ k)$
by (*induct rule*: *typing-trm-typing-cmd.inducts*) *auto*

Shifting mu variables preserves well-typedness.

lemma *liftM-type*:
$\Gamma, \Delta \vdash_T t : T \Longrightarrow \forall k.\ \Gamma, \Delta\langle k{:}U\rangle \vdash_T (\mathit{liftM\text{-}trm}\ t\ k) : T$
$\Gamma, \Delta \vdash_C c \Longrightarrow \forall k.\ \Gamma, \Delta\langle k{:}U\rangle \vdash_C (\mathit{liftM\text{-}cmd}\ c\ k)$
by(*induct rule*: *typing-trm-typing-cmd.inducts*) *auto*

lemma *dropM-type*:
$\Gamma\ ,\ \Delta 1 \vdash_T t : T \Longrightarrow k \notin \mathit{fmv\text{-}trm}\ t\ 0 \Longrightarrow (\forall x.\ x{<}k \longrightarrow \Delta 1\ x = \Delta\ x)$
$\Longrightarrow (\forall x.\ x{>}k \longrightarrow \Delta 1\ x = \Delta\ (x{-}1)) \Longrightarrow \Gamma\ ,\ \Delta \vdash_T \mathit{dropM\text{-}trm}\ t\ k : T$
$\Gamma\ ,\ \Delta 1 \vdash_C c \Longrightarrow k \notin \mathit{fmv\text{-}cmd}\ c\ 0 \Longrightarrow (\forall x.\ x{<}k \longrightarrow \Delta 1\ x = \Delta\ x)$
$\Longrightarrow (\forall x.\ x{>}k \longrightarrow \Delta 1\ x = \Delta\ (x{-}1)) \Longrightarrow \Gamma\ ,\ \Delta \vdash_C \mathit{dropM\text{-}cmd}\ c\ k$
proof (*induct arbitrary*: $k\ \Delta$ **and** $k\ \Delta$ *rule*: *typing-trm-typing-cmd.inducts*)
case (*activate* $\Gamma\ \Delta\ T\ c$)
then show *?case*
apply(*erule-tac* $x = \mathit{Suc}\ k$ **in** *meta-allE*)
apply(*insert fmv-suc*(*1*))
apply(*clarsimp simp add*: *shift-def*)
done
qed *fastforce+*

Lifting λ and μ-variables in contexts preserves contextual typing judgements.

lemma *liftL-ctxt-type*:
assumes $\Gamma, \Delta \vdash_{ctxt} E : T \Leftarrow U$
shows $\forall k.\ \Gamma\langle k{:}T1\rangle, \Delta \vdash_{ctxt} (\mathit{liftL\text{-}ctxt}\ E\ k) : T \Leftarrow U$
using *assms* **by** (*induct rule*: *typing-ctxt.induct*) (*auto simp add*: *liftL-type*)

lemma *liftM-ctxt-type*:
assumes $\Gamma, \Delta \vdash_{ctxt} E : T \Leftarrow U$
shows $\Gamma, \Delta\langle k{:}T1\rangle \vdash_{ctxt} (\mathit{liftM\text{-}ctxt}\ E\ k) : T \Leftarrow U$
using *assms* **by**(*induct rule*: *typing-ctxt.induct*) (*auto simp add*: *liftM-type*)

Substitution lemma for logical substitution.

theorem *subst-type*:
$\Gamma 1, \Delta \vdash_T t : T \Longrightarrow \Gamma, \Delta \vdash_T r : T1 \Longrightarrow \Gamma 1 = \Gamma\langle k{:}T1\rangle \Longrightarrow \Gamma, \Delta \vdash_T t[r/k]^T : T$
$\Gamma 1, \Delta \vdash_C c \Longrightarrow \Gamma, \Delta \vdash_T r : T1 \Longrightarrow \Gamma 1 = \Gamma\langle k{:}T1\rangle \Longrightarrow \Gamma, \Delta \vdash_C c[r/k]^C$
proof(*induct arbitrary*: $\Gamma\ k\ T1\ r$ **and** $\Gamma\ k\ r\ T1$ *rule*: *typing-trm-typing-cmd.inducts*)
case (*lambda* $\Gamma\ T1\ \Delta\ t\ T2$)
then show *?case*

by −(*rotate-tac 2*, *drule liftL-type*(*1*), *fastforce*)
next
case (*activate* Γ Δ *T c*)
then show *?case*
by −(*rotate-tac 2*, *drule liftM-type*(*1*), *fastforce*)
qed *force*+

Substitution lemma for structural substitution. The proof is by induction on the first typing judgement.

lemma *struct-subst-command*:
assumes Γ, Δ $\vdash_T$ *t* : *T* Δ *x* = *T* Γ , Δ′ $\vdash_{ctxt}$ *E* : *U* ⇐ *T1* Δ = Δ′⟨α:*T1*⟩
(Γ , Δ′ $\vdash_{ctxt}$ *E* : *U* ⇐ *T1* ⟹ Δ = Δ′⟨α:*T1*⟩ ⟹ Γ , Δ′⟨β:*U*⟩ $\vdash_T$ *t*[α=β (*liftM-ctxt E* β)]^T : *T*)
shows Γ,(Δ′⟨β:*U*⟩) $\vdash_C$ (<*x*> *t*)[α=β (*liftM-ctxt E* β)]^C
using *assms* **apply** *clarsimp*
apply (*rule conjI*, *force*, *clarsimp*)+
apply (*rule conjI*)
apply (*metis liftM-ctxt-type passivate shift-eq typing-ctxt-correct2*)
apply *force*
done

theorem *struct-subst-type*:
Γ, Δ*1* $\vdash_T$ *t* : *T* ⟹ Γ, Δ $\vdash_{ctxt}$ *E* : *U* ⇐ *T1* ⟹ Δ*1* = Δ⟨α:*T1*⟩ ⟹ Γ, Δ⟨β:*U*⟩ $\vdash_T$ *t*[α=β (*liftM-ctxt E* β)]^T : *T*
Γ, Δ*1* $\vdash_C$ *c* ⟹ Γ, Δ $\vdash_{ctxt}$ *E* : *U* ⇐ *T1* ⟹ Δ*1* = Δ⟨α:*T1*⟩ ⟹ Γ, Δ⟨β:*U*⟩ $\vdash_C$ *c*[α=β (*liftM-ctxt E* β)]^C
proof (*induct arbitrary*: Δ *T1 E U* β α **and** Δ *T1 E U* β α *rule*: *typing-trm-typing-cmd.inducts*)
case (*lambda* Γ *T1* Δ *t T2*)
then show *?case*
by −(*drule liftL-ctxt-type*, *fastforce simp*: *liftLM-comm-ctxt*)
next
case (*activate* Γ Δ *T c*)
then show *?case*
by −(*drule liftM-ctxt-type*, *fastforce simp*: *liftMM-comm-ctxt*)
next
case (*passivate* Γ Δ *t T x*)
then show *?case*
using *struct-subst-command* **by** *force*
qed *fastforce*+

lemma *struct-subst-type-command*: Γ, Δ*1* $\vdash_C$ *c* ⟹ Γ, Δ $\vdash_{ctxt}$ *E* : *U* ⇐ *T1*
⟹ Δ*1* = Δ⟨α:*T1*⟩
⟹ Γ, Δ⟨β:*U*⟩ $\vdash_C$ *c*[α=β (*liftM-ctxt E* β)]^C
using *struct-subst-type* **by** *force*

lemma *dropM-env*:
Γ, $\Delta 1 \vdash_T t[k{=}x\ \Diamond]^T : T \Longrightarrow \Delta 1 = \Delta\langle x{:}(\Delta\ x)\rangle \Longrightarrow \Gamma\ ,\ \Delta \vdash_T$ *dropM-trm* $(t[k{=}x\ \Diamond]^T)\ x : T$
$\Gamma, \Delta 1 \vdash_C c[k{=}x\ \Diamond]^C \Longrightarrow \Delta 1 = \Delta\langle x{:}(\Delta\ x)\rangle \Longrightarrow \Gamma\ , \Delta \vdash_C$ *dropM-cmd* $(c[k{=}x\ \Diamond]^C)\ x$
by (*induct t* **and** *c arbitrary*: $\Gamma\ T\ \Delta 1\ x\ k\ \Delta$ **and** $\Gamma\ T\ \Delta 1\ x\ k\ \Delta$) *fastforce+*

theorem *type-preservation*:
$\Gamma, \Delta \vdash_T t : T \Longrightarrow t \longrightarrow s \Longrightarrow \Gamma, \Delta \vdash_T s : T$
$\Gamma, \Delta \vdash_C c \Longrightarrow c\ {}_C\!\longrightarrow d \Longrightarrow \Gamma, \Delta \vdash_C d$
proof(*induct arbitrary*: *s* **and** *d rule*: *typing-trm-typing-cmd.inducts*)
case (*app* $\Gamma\ \Delta\ t\ T1\ T2\ s\ s'$)
then show *?case*
apply −
apply (*erule redE*; *clarsimp simp*: *subst-type*(*1*))
apply (*subgoal-tac* $\Gamma\ ,\ \Delta \vdash_T$ *ctxt-subst* $(\Diamond \bullet s)$ $(\mu(T1 \to T2) : c) : T2$)
apply (*drule typing-ctxt-correct1*, *clarsimp*)
apply (*subgoal-tac* $\Gamma\ ,\ \Delta\langle 0{:}T2\rangle \vdash_C c[0{=}0\ (liftM\text{-}ctxt\ (\Diamond \bullet s)\ 0)]^C$)
apply (*clarsimp*, *rule struct-subst-type-command*)
apply *force+*
done
next
case (*passivate* $\Gamma\ \Delta\ t\ T\ x$)
then show *?case*
apply *clarsimp*
apply (*erule redE*, *clarsimp*)
apply (*drule struct-subst-type-command* [**where** $\beta = x$])
apply (*fastforce simp*: *dropM-env*(*2*))+
done
qed (*force simp*: *dropM-type*)+

end

1.8 Progress

theory *Progress*
imports *TypePreservation*
begin

We say that a term t is in *weak-head-normal* form when one of the following conditions are met:

1. t is a value,
2. there exists α and v such that $t = \mu : \tau.[\alpha]\ v$ with $\alpha \in fcv(v)$ whenever $\alpha = 0$.

fun (*sequential*) *is-nf* :: *trm* ⇒ *bool* **where**

is-nf (μ *U*: (<β> *v*)) = (*is-val v* $\wedge$ (β = *0* $\longrightarrow$ *0* $\in$ *fmv-trm v 0*)) |
is-nf v = *is-val v*

lemma *progress′*:
Γ, Δ $\vdash_T$ *t* : *T* $\Longrightarrow$ *lambda-closed t* $\Longrightarrow$ ($\forall$ *s*. $\neg$(*t* $\longrightarrow$ *s*)) $\Longrightarrow$ *is-nf t*
Γ, Δ $\vdash_C$ *c* $\Longrightarrow$ *lambda-closedC c* $\Longrightarrow$ ($\forall$ β *t*. *c* = (<β> *t*) $\longrightarrow$ ($\forall$ *d*. $\neg$(*t* $\longrightarrow$ *d*)) $\longrightarrow$ *is-nf t*)
proof (*induct rule*: *typing-trm-typing-cmd.inducts*)
case (*app* Γ Δ *t T1 T2 s*)
then show *?case*
by −(*erule type-arrow-elim*; *force*)
next
case (*activate* Γ Δ *T c*)
then show *?case*
proof(*cases c*)
case (*MVar* α *t*)
then show *?thesis*
using *activate* **by** (*case-tac t*; *force*)
qed
qed *force+*

theorem *progress*:
assumes Γ, Δ $\vdash_T$ *t* : *T lambda-closed t*
shows *is-nf t* $\vee$ ($\exists$ *s*. *t* $\longrightarrow$ *s*)
using *assms* **by** (*fastforce intro*: *progress′*)

end

1.9 Peirce

theory *Peirce*
imports *Types*
begin

As an example of our $\lambda\mu$ formalisation, we show show a $\lambda\mu$-term inhabiting Peirce's Law. The example is due to Parigot [4].

Peirce's law: $((A \to B) \to A) \to A$.

lemma Γ, Δ $\vdash_T$ λ (*A*→*B*)→*A*: (μ *A*:(<*0*>((‘*0*) ° (λ *A*: (μ *B*:(<*1*> (‘*0*)))))))
: ((*A*→*B*)→*A*)→*A*
by *fastforce*

end

References

[1] H. Geuvers, R. Krebbers, and J. McKinna. The $\lambda\mu^{\mathtt{T}}$-calculus. *Annals of Pure and Applied Logic*, 164(6), 2013.

[2] T. Griffin. A formulae-as-types notion of control. In *POPL*, 1990.

[3] C. Matache, V. B. F. Gomes, and D. P. Mulligan. $\lambda\mu$-calculus and muML public Bitbucket repository: https://bitbucket.org/Cristina_Matache/prog-classical-types/, 2017.

[4] M. Parigot. Lambda-Mu-Calculus: An algorithmic interpretation of Classical Natural Deduction. In *LPAR*, 1992.